Basic 1D Pulse Sequences Theory and Practice

NMR Club – 03.09.19

Gaston Courtade, PhD

References

- Bruker TopSpin User Manual (2018) Introduction to Bruker NMR Pulse Programming

  available from your TopSpin installation
  C:*Bruker**TopSpin4.0.6**prog**docu**English**topspin**pdf*

- Lectures by James Keeler http://www-keeler.ch.cam.ac.uk/lectures/

  also available as video lectures https://youtu.be/nM7jQFhrvR0

The basic pulse–acquire experiment



1. The sample is allowed to come to equilibrium.
2. RF power is switched on for long enough to rotate the magnetization through 90° i.e. a 90° pulse is applied.
3. After the RF power is switched off we start to detect the signal which arises from the magnetization as it rotates in the transverse plane.





1. Equilibrium magnetization builds up along the ***z*** axis

$M_z = M_0$

2. The 90° pulse rotates this magnetization into the ***-y*** axis

$M_y = -M_0 \cos(\Omega t)$

$M_x = M_0 \sin(\Omega t)$

3. These magnetizations vary with time.

Integration of these signals gives a spectrum with a peak at frequency **Ω**.



If we acquire the signal right after the pulse, with the receiver at $-\boldsymbol{y}$,

the $\mathbf{M_y}$ signal gives an *absorption* mode spectrum

and the $\mathbf{M_x}$ signal gives a *dispersion* mode spectrum.





absorption



dispersion

The goal is to acquire signals with the same phase. So that through signal averaging after a certain number of scans, **NS**, the signals add to increase the signal-to-noise ratio.

$$S/N \propto \sqrt{NS}$$

Basic pulse programming in TopSpin

```
1  ;zg
2  ;avance-version (12/01/11)
3  ;1D sequence
4  ;
5  ;$CLASS=HighRes
6  ;$DIM=1D
7  ;$TYPE=
8  ;$SUBTYPE=
9  ;$COMMENT=
10
11
12 #include <Avance.incl>
13
14
15 "acqt0=-p1*2/3.1416"
16
17
18 1 ze
19 2 30m
20   d1
21   p1 ph1
22   go=2 ph31
23   30m mc #0 to 2 F0(zd)
24 exit
```

These are comments.
They always start with a colon.

This is an #include statement.
It imports code necessary for the execution of the program.

This is a calculation of parameter `acqt0`,the point t=0 in the FID.
Calculations/variables are always enclosed by double quotes.

This is the bulk of the pulse program.

The basic pulse–acquire experiment in practice




```
18  1 ze
19  2 30m
20    d1
21    p1 ph1
22    go=2 ph31
23    30m mc #0 to 2 F0(zd)
24  exit
25
26
27  ph1=0 2 2 0 1 3 3 1
28  ph31=0 2 2 0 1 3 3 1
```

```
31  ;pl1 : f1 channel - power level for pulse (default)
32  ;p1 : f1 channel -  high power pulse
33  ;d1 : relaxation delay; 1-5 * T1
34  ;ns: 1 * n, total number of scans: NS * TDO
```

`1 ze`

18: starts with a *label* (“1“). They can be reached by loop or branch statements such as `go=label`,
`go to label times n` or `goto label`

The statement `ze` resets the scan counter (displayed during acquisiton) to 0, and enables the execution of dummy scans.

`2 30m`

19: starts with label “2“. It executes a 30 ms delay.

`d1`

20: executes the d1 delay.

`p1 ph1`

21: execute a pulse of length `p1`, power level `pl1` and phase `ph1` on frequency channel `f1`.

`go=2 ph31`

22: execute the pre-scan-delay `de`, execute one data acquisition scan with receiver phase `ph31`, then loop to the pulse program line with label “2“. Repeat this until NS scans have been accumulated.

`30m mc #0 to 2 F0(zd)`

23: execute a 30ms delay, during which data is written to disk.

`exit`

24: finalize execution of the pulse program

Phase cycling

```
27  ph1=0 2 2 0 1 3 3 1
28  ph31=0 2 2 0 1 3 3 1
```

`ph1`=phases for P1
`ph31`=phases for the receiver

0, 1, 2, 3 are multiples of 90°

0 = 0° = x
1 = 90° = y
2 = 180° = -x
3 = 270° = -y

The receiver phase follows the phase of the magnetization.

The receiver phase is advanced in step so as to always be in the same position *relative* to the magnetization. The result is that the lineshape is the same for each repetition of the experiment so that they can all be added together without cancellation.

<u>This is exactly what we require as a FID is time-averaged.</u>





1D ^{13}C sequences

Gated decoupling:

-NOE-buildup during ^{1}H decoupling
(higher S/N but non-quantitative)

-Splitting due to recoupling during acquisition
(extra information: C-H coupling constants)





Inverse-gated decoupling

-No NOE-buildup (lower S/N but quantitative)

-No splitting due to decoupling during acquisition (only chemical shift information: simpler spectra)





http://triton.iqfr.csic.es/guide/nmr/manual/6a_advc.html

1D 13C spectrum of polyM





^{13}C sequence with power-gated decoupling



What 1D ^{13}C experiment is this?

- One signal (only the C-1 is labeled).
- Only chemical shift information
- Is it inverse-gated decoupled?

- During the first decoupling segment with **PL13≈0.5* PL12**, an NOE build-up pulse during the recycle delay is of lower power.
- The second decoupling segment with **PL12** during acquisition is the true decoupling pulse.
- This can be useful to avoid RF heating on salty samples or probes where a higher decoupling power can be problematic.

^{13}C sequence with power-gated decoupling

```
"d11=30m"
"DELTA=d1-100m"
"acqt0=-p1*2/3.1416"

1 ze
  d11 pl12:f2
2 30m do:f2
  10u pl13:f2
  d11 cpd2:f2
  DELTA
  4u do:f2
  10u pl12:f2
  100m cpd2:f2
  p1 ph1
  go=2 ph31
  30m do:f2 pl13:f2 mc #0 to 2 F0(zd)
exit

ph1=0 2 2 0 1 3 3 1
ph31=0 2 2 0 1 3 3 1

;pl1 : f1 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;pl13: f2 channel - power level for second CPD/BB decoupling
;p1 : f1 channel -  high power pulse
;d1 : relaxation delay; 1-5 * T1
;d11: delay for disk I/O                              [30 msec]
;ns: 1 * n, total number of scans: NS * TD0
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence
```

`2 30m do:f2`

22: label «2», during a delay of 30 ms, switch off the decoupler (“decoupler off”) in channel **f2**

`10u pl13:f2`

23: during a delay of 10 µs, set the power level to **PL13** in channel **f2**

`d11 cpd2:f2`

24: during delay **d11** switch on the decoupler in channel **f2**

Exercises

1. Suggest a pulse sequence for the `zg30` experiment (1D proton with a 30° pulse)



2. Which phase cycling scheme would result in total signal cancellation of this experiment?
3. Suggest a pulse sequence for the `zgig` experiment (1D carbon with inverse-gated ^{1}H decoupling)